

BAGIAN I

PENGANTAR PEMROGRAMAN MACRO MINITAB

1. Pendahuluan

Pada umumnya program paket Minitab digunakan dengan cara menulis *command* dalam *Session window* atau dengan memilih menu yang telah tersedia. Selain melalui cara-cara tersebut, kita dapat menggunakan Minitab melalui macro. Macro Minitab merupakan himpunan dari *command* Minitab dan *statement* macro yang disimpan dalam sebuah file, biasanya dalam file text. Macro Minitab dapat digunakan untuk automisasi tugas yang berulang, mengembangkan fungsi-fungsi yang telah tersedia dalam Minitab atau untuk simulasi. Biasanya macro Minitab dibuat dengan tujuan untuk menjawab permasalahan yang spesifik atau lebih kompleks dimana fasilitas tersebut masih belum tersedia secara langsung dalam Minitab.

Terdapat tiga (3) jenis macro Minitab yaitu:


- a. Execs
- b. GLOBAL Macros (simple macro)
- c. LOCAL Macros (advanced macro)

Global macros dan local macros mempunyai fungsi dan kemampuan yang lebih baik daripada pendahulunya yaitu execs. Baik global maupun local macros dapat mengandung *command* Minitab, *statement* macro seperti loop DO dan statement IF untuk mengontor alur macro, serta subroutine. Tetapi local macros memiliki beberapa kelebihan dibanding global macros, misalnya kemampuan untuk menggunakan argument, variabel lokal dan *subcommand* nya sendiri.

2. Global and Local Worksheets

Di dalam Minitab terdapat suatu worksheet, yang disebut global worksheet, berisi semua kolom, konstanta dan matriks. Ini adalah worksheet yang ada pada waktu kita membuka Minitab, di dalam worksheet ini pula global macros bekerja.

Di lain pihak, setiap local macro bekerja pada worksheetnya masing-masing, yang disebut local worksheet. Jadi sebuah local macro akan mempunyai kolom, konstanta dan matriksnya sendiri yang terpisah dari kolom, konstanta dan matriks di global worksheet. Menggunakan variabel-variabel lokal tidak berpengaruh pada global worksheet kecuali variabel yang telah dijalankan pada local macro sebagai argument. Setelah local macro dijalankan, semua variabel tersebut akan hilang.



Gambar 1. Interface Minitab

3. GLOBAL MACROS

3.1 Struktur GLOBAL Macros

Global macro mempunyai struktur sebagai berikut:

```
GMACRO  
  Template  
  Body of the macro  
ENDMACRO
```

Keterangan:

- GMACRO dan ENDMACRO menandai awal dan akhir program macro
- *Template* berisi nama macro. Template harus diawali dengan karakter dan selanjutnya bisa diikuti oleh huruf, angka atau underscore. Huruf kecil atau huruf besar tetap dibaca sama oleh Minitab. Nama template tidak harus selalu sama dengan nama file, tetapi akan lebih memudahkan jika keduanya sama.

Template	File name	Dieksekusi dengan
MyMacro	MYMACRO.MAC	%MYMACRO
Analyze	TEST.MAC	%TEST
Analyze2	TEST2.TXT	%TEST2.TXT

- *Body of the macro* dapat berisi
 - MINITAB commands
 - Control statements
 - Macro statements (seperti IF, THEN, PAUSE, CALL and GOTO)
 - Memanggil global macro lain

3.2 Membuat dan Menyimpan Global macros

Terdapat dua cara untuk membuat global macro:

1. Tuliskan macro yang diinginkan (sesuai struktur global macro) menggunakan *text editor* atau *word processor* seperti notepad, kemudian file macro disimpan dalam format teks (.txt) atau dengan default format file macro Minitab (.mac).
2. Menggunakan minitab session.
 - a. Jalankan beberapa command baik menggunakan menu command atau session command.
 - b. Klik pada Window > Project manager > History
Folder ini menampilkan command-command terakhir yang paling sering dijalankan dalam session
 - c. Blok command yang ingin dimasukkan dalam macro, kemudian Edit > Copy.
 - d. Buka sebarang aplikasi word processing, pilih Edit > Paste.
 - e. Ubah command sebagaimana yang diinginkan, sisipkan tiga baris yang meliputi GMACRO, template dan ENDMACRO.
 - f. Simpan dalam bentuk file teks, contoh:

NAMAFILE.TXT atau "NAMAFILE.MAC"

3.3 Contoh Global Macro

Nama file : ANALYZE.TXT.

GMACRO	Menandai awal global macro
Analyze	Template atau nama dari macro
NAME C1 "Yield" C2 "Chem1" C3 "Ln.Yield" RANDOM 100 C1-C2; UNIFORM 2 10. LET C3=LN('Yield') PRINT C1-C3 DESCRIBE C1-C3 REGRESS C3 2 C1 C2	Body of the macro Membangkitkan bil acak berdistribusi uniform dg batas bawah 2 dan batas atas 10. Statistika Deskriptif Regresi linear
ENDMACRO	Menandai akhir dari global macro

3.4 Menjalankan Global macro

Secara umum untuk menjalankan global macro adalah dengan menuliskan % diikuti nama file macro beserta ekstensi file. Sebelumnya aktifkan Enable Command pada window Session dengan cara klik Editor > Enable Command.

Eksekusi	Ekstensi file	Disimpan pada
%ANALYZE	MAC	Subdirektori macros pada Minitab
%ANALYZE.TXT	bukan MAC	Subdirektori macros pada Minitab
%c:\SALES\ANALYZE	MAC	C:\SALES
%c:\SALES\ANALYZE.TXT	Bukan MAC	C:\SALES
%'nama file.MAC' %'c:\ANALYZE 1.TXT'	Jika pada nama file terdapat spasi	

Contoh: MTB > %c:\analyze.txt

3.5 Control Statement dalam pemrograman macro

- statemen IF : tindakan dilakukan hanya jika beberapa kondisi adalah benar atau salah
- loop DO-ENDDO : suatu tindakan dilakukan selama suatu bilangan tertentu
- WHILE-ENDWHILE : mengulang blok command selama suatu kondisi benar
- CALL dan RETURN : memulai macro yang lain dari suatu macro

4 LOCAL MACROS

Local macros lebih kompleks daripada global macros, sehingga lebih sulit untuk membuat local macro. Tetapi, local macros lebih fleksibel dan mempunyai kemampuan lebih baik daripada global macros. Local macros dapat menggunakan variabel, argument dan subcommand sementara, untuk meningkatkan kapabilitas pemrosesan macro.

4.1 Elemen dan Struktur LOCAL Macros

Local macro mempunyai kemampuan untuk menangani beberapa elemen yang dapat meningkatkan kapabilitas pemrosesan macro. Beberapa elemen dari Local macros:

1. Variabel
2. Argumen
3. Subcommand

Keterangan:

Tipe Variabel	Dideklarasikan dengan	Isi
Subcommand	Tidak dideklarasikan	An <i>implicit constant</i> that has a value of either 1 (if the subcommand was invoked) or 0 (if the subcommand was not invoked)
Text	MCONSTANT	A text constant that contains a text string
Suffixed	MCOLUMN MCONSTANT	A range of columns or constants

Local macro mempunyai struktur sebagai berikut:

MACRO

Template

Declaration statement

Body of the macro

ENDMACRO

Keterangan:

a. Menulis template

Baris pertama dari template berisi nama macro. Dalam satu file dapat berisi lebih dari satu macro. Setiap macro harus mengikuti struktur local macro dan setiap macro harus mempunyai nama yang sifatnya tunggal. Minitab akan menjalankan macro pertama yang ada dalam file. Kita bisa menjalankan macro berikutnya dengan menggunakan statemen CALL diikuti tiap nama template.

Contoh template untuk command dengan argumen:

Template	Cara menjalankan
Trim X Xbar	%TRIM C5 K1

Dalam template di atas, Trim adalah command (dan nama macro), X adalah argument pertama, Xbar adalah argument kedua. Variabel X adalah kolom yang akan dicari datanya oleh macro. Xbar adalah konstanta tempat hasil disimpan.

Contoh template untuk command dengan subcommand

Template	Cara menjalankan
Trim X Xbar; Percent Pct.	%TRIM C1 K1; PERCENT 0.05.

Command TRIM mempunyai argument X dan Xbar. Subcommand-nya adalah Percent. Percent mempunyai argument Pct, yang berisi konstanta.

b. Declaration statements

Semua variabel yang akan digunakan dalam local macro harus dideklarasikan:

- Pendeklarasian variabel :
 - konstanta dengan MCONSTANT
 - kolom dengan MCOLUMN
 - matriks dengan MMATRIX
- Argumen, adalah variabel dalam template
- Deklarasi variabel hanya dapat dibuat antara template dan body of the macro
- Deklarasi statement dapat ditulis beberapa kali tapi untuk variabel yang berbeda, sekali suatu variabel dideklarasikan maka tidak bisa dideklarasikan lagi.

- Command deklarasi (MCOLUMN, MCONSTANT dll) tidak bisa disingkat

Contoh mendeklarasikan variabel:

Misal template adalah TRIM X Xbar.

TRIM adalah nama macro, X dan Xbar adalah variabel yang akan dimasukkan dalam macro. Macro memerlukan declaration statement yang mendefinisikan apakah X dan Xbar adalah konstanta, kolom atau matriks.

4.1 Contoh Local Macros

- | | |
|---|--|
| (1) MACRO | Menandai awal macro |
| (2) TRIM X XBAR;
Percent Pct.
#
TRIM takes one column, X, as input.
It orders the data, trims 5% from each end,
calculates the mean of the remaining data,
and stores it in the constant XBAR.
| Template

Untuk menjalankan macro ini dengan dua argument:
Argument 1: kolom data yang akan di trim
Argument 2: konstanta dimana trimmed mean akan disimpan |
| (3) MCONSTANT N T1 T2 XBAR Pct
MCOLUMN X XSORT XTRIM | Declaration statements.

MCONSTANT mendeklarasikan 4 konstanta: N, T1, T2, dan XBAR untuk digunakan sebagai variabel dalam local macro. XBAR, adalah argument yang berkaitan dengan konstanta yang akan dimasukkan dalam macro ketika macro dijalankan.

MCOLUMN mendeklarasikan tiga kolom (X, XSORT, and XTRIM) yang digunakan sebagai variabel dalam local macro. X adalah argument yang berkaitan dengan kolom yang akan dimasukkan dalam macro ketika macro dijalankan. |
| (4) #
first we calculate the trimming points T1
#and T2
LET N = COUNT(X)
LET T1 = ROUND(N*Pct)
LET T2 = N-T1+1
next we check for the case when T1 = 0 | Body of the macro |

```
#and nothing is trimmed
IF T1 = 0
  LET XTRIM = X
# otherwise, we sort X, trim the ends and
#calculate the mean
ELSE
  LET XSORT = SORT(X)
  COPY XSORT XTRIM;
  OMIT 1:T1 T2:N.
ENDIF
LET XBAR = MEAN(XTRIM)
Print xbar
```

(5) ENDMACRO

Menandai akhir dari macro

Contoh cara menjalankan:

```
%d:\TRIM C1 K1;
Percent 0.05.
```

4.3 Control Statement

4.3.1 IF, ELSEIF, ELSE, ENDIF

IF	ekspresi logika (blok commands minitab dan statement macro)
ELSEIF	ekspresi logika (blok commands minitab dan statement macro)
ELSE	ekspresi logika (blok commands minitab dan statement macro)
ENDIF	

Untuk ekspresi logika dapat menggunakan operator Boolean berikut

=	atau	EQ	Sama dengan
~=	atau	NE	Tidak sama dengan
<	atau	LT	Kurang dari
>	atau	GT	Lebih dari
<=	atau	LE	Kurang dari atau sama dengan
>=	atau	GE	Lebih dari atau sama dengan
&	atau	AND	
	atau	OR	
~	atau	NOT	

Contoh:

```
GMACRO
SMALL
#
# Ambil data di C1-C3. Cari kolom yang mempunyai nilai mean terkecil
# kemudian cetak hasilnya. Jika mean ketiga kolom sama, maka tidak ada
# kolom yang mempunyai mean terkecil, kemudian cetak sebuah pesan.
#
LET K1=MEAN(C1)
LET K2=MEAN(C2)
LET K3=MEAN(C3)
IF K1 < K2 AND K1 < K3
    PRINT K1 C1
ELSEIF K2 < K1 AND K2 < K3
    PRINT K2 C2
ELSEIF K3 < K1 AND K3 < K2
    PRINT K3 C3
ELSE
    NOTE Note: tidak ada nilai mean terkecil
    PRINT K1 K2 K3
ENDIF
ENDMACRO
```

4.3.2 DO, ENDDO

```
DO          K = suatu bilangan
           (blok commands minitab dan statement macro)
ENDDO
```

Contoh:

```
MACRO
MOVAVE X Y
#
# Calculates the simple moving average of the data in X and
# stores the answer in Y.
#
MCONSTANT N I
MCOLUMN X Y
LET N = COUNT(X)
LET Y(1) = '*'
LET Y(2) = '*'
DO I = 3 : N
    LET Y(I) = (X(I) + X(I-1) + X(I-2))/3
ENDDO
ENDMACRO
```

4.3.3 WHILE, ENDWHILE

WHILE ekspresi logika (blok commands minitab dan statement macro) ENDWHILE

Contoh:

MACRO
CTHWHILE X Y

Calculates the simple moving average of the data in X and
stores the answer in Y.

MCONSTANT N I RATA
MCOLUMN X Y
LET N = COUNT(X)
LET RATA=MEAN(X)
PRINT RATA

LET I = 3
WHILE I <= N
 LET Y(I) = (X(I) + X(I-1) + X(I-2))/3
 LET I = I + 1
ENDWHILE
ENDMACRO

4.3.4 GOTO, MLABEL

GOTO number (commands minitab dan statement macro lainnya) MLABEL number

Contoh:

GMACRO
NOMISS
LET K90 = COUNT(C1)
Print k90
DO K91 = 1: K90
 IF C1(K91) = '*'
 GOTO 5
 ENDIF
ENDDO
MLABEL 5
DELETE K91:K90 C1
ENDMACRO

4.4 MENJALANKAN MACRO DARI DALAM MACRO

Contoh:

File utama ANALIS.TXT menentukan berapa banyak pengamatan dalam himpunan data. Jika banyak pengamatan < 5 maka jalankan file TOOSMALL.TXT, jika sebaliknya maka jalankan OK.TXT.

ANALIS.TXT

```
GMACRO
ANALIS
LET K90 = COUNT(C1)
IF K90 < 5
    %C:\TOOSMALL.TXT
ELSE
    %C:\OK.TXT
ENDIF
ENDMACRO
```

TOOSMALL.TXT

```
GMACRO
TOOSMALL
NOTE Data kurang dari 5 pengamatan
PRINT C1-C3
ENDMACRO
```

OK.TXT

```
GMACRO
OK
PRINT C1-C3
DESCRIBE C1-C3
LET C5=LN(C1)
REGRESS C5 2 C2 C3
ENDMACRO
```

4.4.1 CALL, RETURN

Didalam sebuah file dapat terdiri lebih dari satu macro. Untuk mengontrol macro yang ada digunakan CALL dan RETURN

CALL template RETURN

Contoh:

ANALIS.TXT

```
GMACRO  
ANALIS  
LET K90 = COUNT(C1)  
IF K90 < 5  
    CALL TOOSMALL  
    PRINT C1-C3  
ELSE  
    CALL OK  
    PRINT C1-C3 C5  
ENDIF  
ENDMACRO
```

```
GMACRO  
TOOSMALL  
NOTE Data kurang dari 5 pengamatan  
RETURN  
ENDMACRO
```

```
GMACRO  
OK  
PRINT C1-C3  
DESCRIBE C1-C3  
LET C5=LN(C1)  
REGRESS C5 2 C2 C3  
RETURN  
ENDMACRO
```

Soal:

1. Buatlah program untuk menghitung rata-rata menggunakan fasilitas looping di macro Minitab.
2. Buatlah program untuk menghitung nilai median dari data.
3. Buatlah program untuk menghitung nilai Min, Max, Med, Q1 dan Q3 (tanpa fasilitas sort di Minitab)